
Constants Sometimes we want to solve a Gröbner basis for situations where there are constants involved.

We cannot simply consider these constants as further variables (e.g. one cannot solve for these.) Instead,

suppose that we have constants c1, . . . ,ck and variables x1, . . . ,xn.

We first form the polynomial ring in the constants, C := k[c1, . . . ,ck]. This is an integral domain, we therefore

(Theorem 2.25 in the book) can from the field of fractions (analogous to how the rationals are formed as

fractions of integers) F = Frac(C) = k(c1, . . . ,ck)1.

We then set R = F [x1, . . . ,xn] = k(c1, . . . ,ck)[x1, . . . ,xn] and work in this polynomial ring. (The only difficulty

arises once one specializes the constants: It is possible that denominators evaluate back to zero.)

For example, suppose we want to solve the equations

x2y+ y+a = y2x+ x = 0

In GAP we first create the polynomial ring for the constants:

gap> C:=PolynomialRing(Rationals,["a"]);
Rationals[a]
gap> a:=IndeterminatesOfPolynomialRing(S)[1];
a

(There is no need to create a separate fraction field. Now we define the new variables x,y to have coefficients

from C:

gap> x:=X(C,"x");;
#I You are creating a polynomial *over* a polynomial ring (i.e. in an
#I iterated polynomial ring). Are you sure you want to do this?
#I If not, the first argument should be the base ring, not a polynomial ring
#I Set ITER_POLY_WARN:=false; to remove this warning.
gap> y:=X(C,"y");;

As one sees, GAP warns that these variables are defined over a ring which has already variables, but we can

ignore this warning2, or set ITER_POLY_WARN:=false; to turn it off.

Now we can form polynomials and calculate a Gröbner basis:

gap> f:=x^2*y+y+a;;
gap> g:=y^2*x+x;;
gap> ReducedGroebnerBasis([f,g],MonomialLexOrdering());
[ y^3+a*y^2+y+a, x*y^2+x, x^2-y^2-a*y ]

We thus need that y3 +ay2 + y+a = 0, and then can solve this for y and substitute back.

1The round parentheses denote that we permit also division
2The reason for the warning is that this was a frequent error of users when defining variables and polynomial rings. Variables

were taken accidentally not from the polynomial ring, but created over the polynomial ring.



The Nullstellensatz and the Radical In applications (see below) we sometimes want to check not whether

g ∈ I C R, but whether g has “the same roots” as I. (I.e. we want that g(x1, . . . ,xn) = 0 if and only if

f (x1, . . . ,xn) = 0 for all f ∈ I.) Clearly this can hold for polynomials not in I, for example consider I =〈
x2
〉
CQ[x], then g(x) = x 6∈ I, but has the same roots.

An important theorem from algebraic geometry show that over C, such a power condition is all that can

happen3:

Theorem (Hilbert’s Nullstellensatz): Let R = C[x1, . . . ,xn] and ICR. If g∈R such that g(x1, . . . ,xn)=
0 if and only if f (x1, . . . ,xn) = 0 for all f ∈ I, then there exists an integer m, such that gm ∈ I.

The set {g ∈ R | ∃m : gm ∈ I} is called the Radical of I.

There is a neat way how one can test this property, without having to try out possible m: Suppose R =
k[x1, . . . ,xn] and I = 〈 f1, . . . , fm〉CR. Let g ∈ R. Then there exists a natural number m such that gm ∈ I if and

only if (we introduce an auxiliary variable y)

1 ∈ 〈 f1, . . . , fm,1− yg〉=: Ĩ C k[x1, . . . ,xn,y]

The reason is easy: If 1 ∈ Ĩ, we can write

1 = ∑
i

p(x1, . . . ,xn,y) · fi(x1, . . . ,xn)+q(x1, . . . ,xn,y) · (1− yg)

We now set y = 1/g (formally in the fraction field) and multiply with a sufficient high power (exponent m)

of g, to clean out all g in the denominators. We obtain

gm = ∑
i

gm · p(x1, . . . ,xn,1/gm)︸ ︷︷ ︸
∈R

· fi(x1, . . . ,xn)+gm ·q(x1, . . . ,xn,y) · (1− y/y)︸ ︷︷ ︸
=0

∈ I.

Vice versa, if gm ∈ I, then

1 = ymgm +(1− ymgm) = ymgm︸ ︷︷ ︸
∈I⊂Ĩ

+(1− yg)︸ ︷︷ ︸
∈Ĩ

(1+ yg+ · · ·+ ym−1gm−1) ∈ Ĩ

This property can be tested easily, as 1 ∈ Ĩ⇔ Ĩ = k[x1, . . . ,xn,y], which is the case only if a Gröbner basis

for Ĩ contains a constant.

We will see an application of this below.

Proving Theorems from Geometry Suppose we describe points in the plane by their (x,y)-coordinates.

We then can describe many geometric properties by polynomial equations:

Theorem: Suppose that A = (xa,ya), B = (xb,yb), C = (xc,yc) and D = (xd ,yd) are points in the plane. Then

the following geometric properties are described by polynomial equations:

AB is parallel to CD:
yb− ya

xb− xa

=
yd− yc

xd− xc

, which implies (yb− ya)(xd− xc) = (yd− yc)(xb− xa).

AB⊥CD: (xb− xa)(xd− xc)+(yb− ya)(yd− yc) = 0.

|AB|= |CD|: (xb− xa)2 +(yb− ya)2 = (xd− xc)2 +(yd− yc)2.

C lies on the circle with center A though the point B: |AC|= |AB|.
3Actually, one does not need C, but only that the field is algebraically closed



C is the midpoint of AB: xc =
1

2
(xb− xa), yc = 1

2
(yb = ya).

C is collinear with AB: self

BD bisects ∠ABC: self

A theorem from geometry now sets up as prerequisites some points (and their relation with circles and

lines) and claims (the conclusion) that this setup implies other conditions. If we suppose that (xi,yi) are

the coordinates of all the points involved, the prerequisites thus are described by a set of polynomials f j ∈
Q[x1, . . . ,xn,y1, . . . ,yn] =: R. A conclusion similarly corresponds to a polynomials g ∈ R.

The statement of the geometric theorem now says that whenever the {(xi,yi)} are points fulfilling the prereq-

uisites (i.e. f j(x1, . . . ,xn,y1, . . . ,yn)= 0), then also the conclusion holds for these points (i.e. g(x1, . . . ,xn,y1, . . . ,yn)=
0).

This is exactly the condition we studied in the previous section and translated to gm ∈ 〈 f1, . . . , fm〉, which

we can test.

(Caveat: This is formally over C, but we typically want real coordinates. There is some extra subtlety in

practice.)

For example, consider the theorem of THALES: Any triangle suspended under a half-circle has a right angle.
We assume (after rotation and scaling) that A = (−1,0) and B = (1,0) and set

C = (x,y) with variable coordinates. The fact that C is on the circle (whose

origin is (0,0) and radius is 1) is specified by the polynomial

f = x2 + y2−1 = 0 A

C

B

The right angle at C means that AC ⊥ BC. We encode this as

g = (x− (−1))+(1− x)+ y(−y) =−x2− y2 +1 = 0.

Clearly g is implied by f .

Apollonius’ theorem We want to use this approach to prove a classical geometrical theorem (it is a special

case of the “Nine point” or “Feuerbach circle” theorem):

Circle Theorem of APOLLONIUS: Suppose that

ABC is a right angled triangle with right angle at A.

The midpoints of the three sides, and the foot of the

altitude drawn from A onto BC all lie on a circle.

A

B C

H

M
BC

M
ACM

AB

To translate this theorem into equations, we choose coordinates for the points A, B, C. For simplicity we set

(translation) A = (0,0) and B = (b,0) and C = (0,c), where b and c are constants. (That is, the calculation

takes place in a polynomial ring not over the rationals, but over the quotient field of Q[b,c].) Suppose that

MAB = (x1,0), MAC = (0,x2) and MBC = (x3,x4). We get the equations

f1 = 2x1−b = 0

f2 = 2x2− c = 0

f3 = 2x3−b = 0

f4 = 2x4− c = 0



Next assume that H = (x5,x6). Then AH ⊥ BC yields

f5 = x5b− x6c = 0

Because H lies on BC, we get

f6 = x5c+ x6b−bc = 0

To describe the circle, assume that the middle point is O = (x7,x8). The circle property then means that

|MABO|= |MBCO|= |MACO| which gives the conditions

f7 = (x1− x7)2 +(0− x8)2− (x3− x7)2− (x4− x8)2 = 0

f8 = (x1− x7)2 + x2
8− x2

7− (x8− x2)2 = 0

The conclusion is that |HO|= |MABO|, which translates to

g = (x5− x7)2 +(x6− x8)2− (x1− x7)2− x2
8 = 0.

We now want to show that there is an m, such that gm ∈ 〈 f1, . . . , f8〉. In GAP, we first define a ring for the

two constants, and assign the constants. We also define variables over this ring.

R:=PolynomialRing(Rationals,["b","c"]);
ind:=IndeterminatesOfPolynomialRing(R);b:=ind[1];c:=ind[2];
x1:=X(R,"x1"); ... x8:=X(R,"x8");

We define the ideal generators fi as well as g:

f1:=2*x1-b;
f2:=2*x2-c;
f3:=2*x3-b;
f4:=2*x4-c;
f5:=x5*b-x6*c;
f6:=x5*c+x6*b-b*c;
f7:=(x1-x7)^2+(0-x8)^2-(x3-x7)^2-(x4-x8)^2;
f8:=(x1-x7)^2+x8^2-x7^2-(x8-x2)^2;
g:=(x5-x7)^2+(x6-x8)^2-(x1-x7)^2-x8^2;

For the test whether gm ∈ I, we need an auxiliary variable y. Then we can generate a Gröbner basis for Ĩ:

order:=MonomialLexOrdering();
bas:=ReducedGroebnerBasis([f1,f2,f3,f4,f5,f6,f7,f8,1-y*g],order);

The basis returned is (1), which means that indeed gm ∈ I.

If we wanted to know for which m, we can test membership in I itself, and find that already g ∈ I:

gap> bas:=ReducedGroebnerBasis([f1,f2,f3,f4,f5,f6,f7,f8],order);
[ x8-1/4*c, x7-1/4*b, x6+(-b^2*c/(b^2+c^2)), x5+(-b*c^2/(b^2+c^2)),
x4-1/2*c, x3-1/2*b, x2-1/2*c, x1-1/2*b ]

gap> PolynomialReducedRemainder(g,bas,order);
0


